



CernerMillennium: Discern Explorer 2/3

Exercise Answers



34TRGINT000008

Owner: Product Specialist Team Lead

Reviewed By: Mike Schweder

Approved By: Bob Ross

Answers to Exercises

Exercise 1

```
DROP PROGRAM explorer_23_exercise_1 GO
CREATE PROGRAM explorer_23_exercise_1

!prompts
PROMPT "Output to File/Printer/MINE" = MINE,
!challenge steps
    "Enter A Last Name" = "*",
    "Enter A First Name" = "*"

!report prolog
;Request HNAM sign-on when executed from CCL on host
IF (VALIDATE(IsOdbc, 0) = 0) EXECUTE CCLSECLOGIN ENDIF

!set maximum execute time for query/report
SET MaxSecs = 0
IF (VALIDATE(IsOdbc, 0) = 1) SET MaxSecs = 15 ENDIF

!fields
SELECT   INTO $1
        P.NAME_FULL_FORMATTED,
        E.ENCNTR_ID,
        O.ORDER_MNEMONIC,
        PA.ALIAS

!tables
FROM    PERSON P,
        ENCOUNTER E,
        ORDERS O,
        PERSON_ALIAS PA

!qualifications
PLAN P WHERE P.PERSON_ID > 0
!challenge steps
        AND P.NAME_LAST_KEY = PATSTRING(CNVTUPPER($2))
        AND P.NAME_FIRST_KEY = PATSTRING(CNVTUPPER($3))
/* note if using the prompt builder to convert what the user enters to
uppercase you would not need to use the patstring() and cnvtupper() functions
in the qualifications on p.name_last_key or p.name_first_key */
JOIN E WHERE E.PERSON_ID = P.PERSON_ID
JOIN O WHERE O.ENCNTR_ID = E.ENCNTR_ID
JOIN PA WHERE PA.PERSON_ID = P.PERSON_ID

!control options
WITH MAXREC = 1000, TIME = VALUE( MaxSecs ), NOHEADING, FORMAT= VARIABLE
END
GO
```

34TRGINT000008

Owner: Product Specialist Team Lead

Reviewed By: Mike Schweder

Approved By: Bob Ross

Exercise 2

```
DROP PROGRAM explorer23_class_ex2 GO
CREATE PROGRAM explorer23_class_ex2

!prompts
PROMPT "Output to File/Printer/MINE" = MINE

!report prolog
;Request HNAM sign-on when executed from CCL on host
IF (VALIDATE(IsOdbc, 0) = 0) EXECUTE CCLSECLOGIN ENDIF

!challenge steps
DECLARE SSN_VAR = F8
SET SSN_VAR = UAR_GET_CODE_BY("MEANING", 4, "SSN")

!set maximum execute time for query/report
SET MaxSecs = 0
IF (VALIDATE(IsOdbc, 0) = 1) SET MaxSecs = 15 ENDIF

!fields
SELECT   INTO $1
  P.NAME_FULL_FORMATTED,
  PA.ALIAS,
  PA_PERSON_ALIAS_TYPE_DISP =UAR_GET_CODE_DISPLAY(PA.PERSON_ALIAS_TYPE_CD),
  PA_PERSON_ALIAS_TYPE_CDF = UAR_GET_CODE_MEANING( PA.PERSON_ALIAS_TYPE_CD )

!tables
FROM PERSON P,
      PERSON_ALIAS PA

!qualifications
PLAN P
JOIN PA WHERE OUTERJOIN(P.PERSON_ID) = PA.PERSON_ID
!challenge step
      AND PA.PERSON_ALIAS_TYPE_CD = OUTERJOIN(SSN_VAR)

!control options
WITH MAXREC = 100, TIME = VALUE( MaxSecs ), FORMAT, SKIPREPORT= 1
END
GO
```

Exercise 3

```
DROP PROGRAM explorer23_class_ex3 GO
CREATE PROGRAM explorer23_class_ex3
```

```
!prompts
PROMPT "Output to File/Printer/MINE" = MINE
```

```
!report prolog
;Request HNAM sign-on when executed from CCL on host
IF (VALIDATE(IsOdbc, 0) = 0) EXECUTE CCLSECLOGIN ENDIF
```

```
!set maximum execute time for query/report
```

34TRGINT000008

Owner: Product Specialist Team Lead

Reviewed By: Mike Schweder

Approved By: Bob Ross

```
SET MaxSecs = 0
IF (VALIDATE(IsOdbc, 0) = 1) SET MaxSecs = 15 ENDIF

!fields
SELECT   INTO $1
        P.NAME_FULL_FORMATTED,
        O.ORDER_MNEMONIC,
        OC.LONG_TEXT_ID,
        !challenge step
        LONG_TEXT = SUBSTRING( 1, 100, L.LONG_TEXT )

!tables
FROM    PERSON   P,
        ORDERS  O,
        ORDER_COMMENT OC,
        !challenge step
        LONG_TEXT L

!qualifications
PLAN P WHERE P.PERSON_ID > 0
JOIN O WHERE P.PERSON_ID = O.PERSON_ID AND O.ORDER_ID > 0
JOIN OC WHERE outerjoin( O.ORDER_ID ) = OC.ORDER_ID
!challenge step
JOIN L WHERE outerjoin(OC.LONG_TEXT_ID) = L.LONG_TEXT_ID

!control options
WITH MAXREC = 100, TIME = VALUE( MaxSecs ), FORMAT, SKIPREPORT= 1

END
GO
```

Exercise 4

```
DROP PROGRAM explorer_23_ex_4 GO
CREATE PROGRAM explorer_23_ex_4

!prompts
PROMPT "Output to File/Printer/MINE " = MINE

!report prolog
;Request HNAM sign-on when executed from CCL on host
IF (VALIDATE(IsOdbc, 0) = 0) EXECUTE CCLSECLOGIN  ENDIF

SET MaxSecs = 0
IF (VALIDATE(IsOdbc, 0) = 1) SET MaxSecs = 15 ENDIF
!fields
SELECT INTO $1
        NAME = SUBSTRING(1,20,P.NAME_FULL_FORMATTED),
        E.ENCNTR_TYPE_CLASS_CD,
        E_ENCNTR_TYPE_CLASS_CDF = UAR_GET_CODE_MEANING( E.ENCNTR_TYPE_CLASS_CD )

FROM
        PERSON   P,
        ENCOUNTER E,
```

34TRGINT000008

Owner: Product Specialist Team Lead

Reviewed By: Mike Schweder

Approved By: Bob Ross

ORDERS O

```
PLAN P WHERE P.PERSON_ID >0
JOIN E WHERE E.PERSON_ID = P.PERSON_ID
JOIN O WHERE O.ENCNTR_ID = OUTERJOIN(E.ENCNTR_ID) AND O.ENCNTR_ID = NULL

!control options

WITH MAXREC = 100 , TIME = VALUE( MaxSecs ), FORMAT, SEPARATOR = " "
END
GO
```

Exercise 5

```
DROP PROGRAM explorer_23_ex_5 GO
CREATE PROGRAM explorer_23_ex_5

!prompts
PROMPT "Output to File/Printer/MINE" = MINE

!report prolog
;Request HNAM sign-on when executed from CCL on host
IF (VALIDATE(IsOdbc, 0) = 0) EXECUTE CCLSECLOGIN ENDIF

DECLARE MRN_VAR = F8
SET MRN_VAR = UAR_GET_CODE_BY("MEANING", 4, "MRN")

DECLARE CMRN_VAR = F8
SET CMRN_VAR = UAR_GET_CODE_BY("MEANING", 4, "CMRN")

!set maximum execute time for query/report
SET MaxSecs = 0
IF (VALIDATE(IsOdbc, 0) = 1) SET MaxSecs = 15 ENDIF
!fields
SELECT   INTO $1
        P.PERSON_ID,
        P.NAME_LAST,
        A.ADDRESS_ID,
        ADDR = SUBSTRING(1,20,A.STREET_ADDR)

!tables
FROM    PERSON P,
        ADDRESS A

!qualifications
PLAN P WHERE P.PERSON_ID>0 AND
          NOT EXISTS (SELECT PA.PERSON_ID FROM PERSON_ALIAS PA
                      WHERE PA.PERSON_ID = P.PERSON_ID AND
                        PA.PERSON_ALIAS_TYPE_CD IN (MRN_VAR, CMRN_VAR))
JOIN A WHERE A.PARENT_ENTITY_ID = P.PERSON_ID AND
          A.PARENT_ENTITY_NAME = "PERSON"

!control options
```

34TRGINT000008

Owner: Product Specialist Team Lead

Reviewed By: Mike Schweder

Approved By: Bob Ross

```
WITH MAXREC = 100, TIME = VALUE( MaxSecs ), FORMAT, SEPARATOR = "
", SKIPREPORT= 1
END
GO
```

Exercise 6

```
DROP PROGRAM explorer_23_ex_6 GO
CREATE PROGRAM explorer_23_ex_6

!prompts
PROMPT "Output to File/Printer/MINE" = MINE

!report prolog
;Request HNAM sign-on when executed from CCL on host
IF (VALIDATE(IsOdbc, 0) = 0) EXECUTE CCLSECLOGIN ENDIF

!set maximum execute time for query/report
SET MaxSecs = 0
IF (VALIDATE(IsOdbc, 0) = 1) SET MaxSecs = 15 ENDIF

!fields
SELECT   INTO $1
        P.NAME_FULL_FORMATTED,
        P_SEX_CDF = UAR_GET_CODE_MEANING( P.SEX_CD ),
        P.SEX_CD,
        P.UPDT_DT_TM,
        DATE_COMPARE = datatetimecmp( cnvtdatetime( curdate,0), P.UPDT_DT_TM )

!tables
FROM    PERSON  P

!qualifications
WHERE   P.SEX_CD > 0

!Sorting
ORDER BY      P_SEX_CDF

Head Report
    ROW 1 COL 53 "Class Exercise 6 for Explorer 23"
    ROW 2 COL 9 "Date :"
    ROW 2 COL 17 curdate
    ROW 3 COL 9 "Time :"
    ROW 3 COL 17 curtime
    ROW + 2

Head P_SEX_CDF
    COL 13 "SEX ="
    COL 21 P.SEX_CD
    COL 37 P_SEX_CDF
    ROW + 2

Detail
    NAME_FULL_FORMATTED1 = SUBSTRING( 1, 20, P.NAME_FULL_FORMATTED ),
    COL 5 NAME_FULL_FORMATTED1
```

34TRGINT000008

Owner: Product Specialist Team Lead

Reviewed By: Mike Schweder

Approved By: Bob Ross

```
COL 33  DATE_COMPARE
COL 53  P.UPDT_DT_TM
COL 65  IF ( DATE_COMPARE >365 )
          "That was over a year ago!"
ENDIF
ROW + 1

Foot P_SEX_CDF
ROW + 1
COL 5  "Total number of people with this sex code:"
COL 53  count( P.SEX_CD) ";L"
ROW + 1
COL 5  "Average number of days since the record was updated:"
COL 65  avg( DATE_COMPARE where p.updt_cnt >0) ";L"
ROW + 1
COL 5  "The oldest update was made:"
COL 33  max( DATE_COMPARE)
COL 53  "days ago."
ROW + 1
COL 5  "The newest update made:"
COL 29  min( DATE_COMPARE)
COL 49  "days ago."
ROW + 2

Foot Report
ROW + 1
COL 13  "Total number of records for this report:"
COL 53  count( P.SEX_CD) ";L"
ROW + 1
COL 13  "The average number of days since this report was updated:"
COL 69  avg( DATE_COMPARE where p.updt_cnt >0) ";L"
ROW + 1
COL 13  "The oldest update was made:"
COL 41  max( DATE_COMPARE)
COL 65  "days ago."
ROW + 1
COL 13  "The newest update was made:"
COL 41  min( DATE_COMPARE)
COL 65  "days ago."

!control options
WITH MAXREC = 100, TIME = VALUE( MaxSecs ), NOHEADING, FORMAT= VARIABLE
END
GO
```

Exercise 7

No new source code is created for exercise 7.

The following answers are for HNAM Explorer 3

Exercise 8

```
DROP PROGRAM explorer_23_ex9 GO
```

34TRGINT000008

Owner: Product Specialist Team Lead

Reviewed By: Mike Schweder

Approved By: Bob Ross

```
CREATE PROGRAM explorer23_class_ex11

!prompts
PROMPT "Output to File/Printer/MINE" = MINE

!report prolog
;Request HNAM sign-on when executed from CCL on host
IF (VALIDATE(IsOdbc, 0) = 0) EXECUTE CCLSECLOGIN ENDIF

;if necessary free the record structure
free record person

;define the record structure
record person (
    1 qual [*]
2 person_id = f8
2 name = vc )

!set maximum execute time for query/report
SET MaxSecs = 0
IF (VALIDATE(IsOdbc, 0) = 1) SET MaxSecs = 15 ENDIF

!fields
SELECT   INTO $1
        P.PERSON_ID,
        NAME = P.NAME_FULL_FORMATTED

!tables
FROM    PERSON P

!qualifications
WHERE   P.PERSON_ID > 0

Head Report
    CNT = 0
    CNTX = 0

Detail
    CNT = CNT +1
    IF ( MOD(CNT,10) = 1)
        STAT = ALTERLIST(PERSON->QUAL, CNT +9)
    ENDIF
    PERSON->QUAL[CNT].PERSON_ID = P.PERSON_ID
    PERSON->QUAL[CNT].NAME = NAME

Foot Report
    STAT = ALTERLIST(PERSON->QUAL, CNT)
    FOR(CNTX = 1 TO CNT)
        COL 10
        PERSON->QUAL[CNTR].PERSON_ID
        COL +2
        PERSON->QUAL[CNTR].NAME
        ROW +1
    ENDFOR

!control options
```

Cerner Corporation

Information

Confidential

©Cerner Corporation. All rights reserved. This document contains confidential information which may not be reproduced or transmitted without the express written consent of Cerner.

34TRGINT000008

Owner: Product Specialist Team Lead

Reviewed By: Mike Schweder

Approved By: Bob Ross

```
WITH MAXREC = 100, TIME = VALUE( MaxSecs ), NOHEADING, FORMAT= VARIABLE  
END  
GO
```

Exercise 9

```
DROP PROGRAM explorer_23_ex_9 GO  
CREATE PROGRAM explorer_23_ex_9  
  
!prompts  
PROMPT "Output to File/Printer/MINE" = MINE  
  
!report prolog  
;Request HNAM sign-on when executed from CCL on host  
IF (VALIDATE(IsOdbc, 0) = 0) EXECUTE CCLSECLOGIN ENDIF  
  
; Defining record structure  
free record person  
record person(  
    1 list [*]  
        2 person_id = f8  
        2 street      = c50 )  
  
select into "NL;"  
    a.parent_entity_id,  
    a.street_addr  
from address a  
where a.parent_entity_name = "PERSON" and  
    a.street_addr > "1"  
  
head report  
    cnt = 0  
  
detail  
    ; Allocate space in memory  
    cnt = cnt +1  
    if(mod(cnt,10) = 1)  
        stat = alterlist(person->list, cnt+9)  
    endif  
    ; Loading record structure  
    person->list[cnt].person_id = a.parent_entity_id  
    person->list[cnt].street = substring(1,50,a.street_addr)  
  
foot report  
    stat = alterlist(person->list, cnt)  
with maxrec = 200  
  
!set maximum execute time for query/report  
SET MaxSecs = 0  
IF (VALIDATE(IsOdbc, 0) = 1) SET MaxSecs = 15 ENDIF  
!fields  
SELECT   INTO $1  
    P.NAME_FULL_FORMATTED  
  
!tables
```

34TRGINT000008

Owner: Product Specialist Team Lead

Reviewed By: Mike Schweder

Approved By: Bob Ross

```
FROM PERSON P,
      ( DUMMYT D1 WITH SEQ = VALUE(SIZE(PERSON->LIST,5)) )

!qualifications
PLAN D1
JOIN P WHERE P.PERSON_ID = PERSON->LIST[D1.SEQ].PERSON_ID

Detail
      NAME_FULL_FORMATTED1 = SUBSTRING( 1, 50, P.NAME_FULL_FORMATTED ),
      COL 5 NAME_FULL_FORMATTED1
      COL +2 PERSON->LIST[D1.SEQ].PERSON_ID
      COL +2 PERSON->LIST[D1.SEQ].STREET
      ROW + 1

!control options
WITH MAXREC = 100, TIME = VALUE( MaxSecs ), NOHEADING, FORMAT= VARIABLE

END
GO
```

Exercise 10

```
DROP PROGRAM explorer_23_ex_10 GO
CREATE PROGRAM explorer_23_ex_10

!prompts
PROMPT "Output to File/Printer/MINE" = MINE

!report prolog
;Request HNAM sign-on when executed from CCL on host
IF (VALIDATE(IsOdbc, 0) = 0) EXECUTE CCLSECLOGIN ENDIF

free record per_ord

record per_ord (
  1 name = c25
  1 orders[*]
    2 id   = f8
    2 mne = c25)

!set maximum execute time for query/report
SET MaxSecs = 0
IF (VALIDATE(IsOdbc, 0) = 1) SET MaxSecs = 60 ENDIF

!fields
SELECT   INTO $1
  P.PERSON_ID,
  P.NAME_FULL_FORMATTED,
  O.ORDER_ID,
  O.ORDER_MNEMONIC

!tables
FROM PERSON P,
      ORDERS O

!qualifications
```

Cerner Corporation

Confidential

Information

©Cerner Corporation. All rights reserved. This document contains confidential information which may not be reproduced or transmitted without the express written consent of Cerner.

34TRGINT000008

Owner: Product Specialist Team Lead

Reviewed By: Mike Schweder

Approved By: Bob Ross

```
plan p where p.person_id = 210
JOIN o WHERE p.person_id = o.order_id

!Sorting
ORDER BY P.PERSON_ID,
O.ORDER_ID

Head Report
cnt = 0
cntx = 0

Head P.PERSON_ID
per_ord->name = substring(1,25, p.name_full_formatted)
ROW + 0

Detail
cnt = cnt + 1
if(mod(cnt,10) = 1)
stat = alterlist(per_ord->orders,cnt+9)
endif
per_ord->orders[cnt]->id = o.order_id
per_ord->orders[cnt]->mne = substring(1,25,o.order_mnemonic)

Foot P.PERSON_ID
ROW + 0

Foot Report
stat = alterlist(per_ord->orders,cnt)
col 0 per_ord->name
row +1
for (cntx = 1 to cnt)
col 10 per_ord->orders[cntx]->id
col +2 per_ord->orders[cntx]->mne
row +1
endfor

!control options
WITH MAXREC = 20, TIME = VALUE( MaxSecs ), NOHEADING, FORMAT= VARIABLE

END
GO
```

Exercise 11

```
DROP PROGRAM explorer_23_ex_11 GO
CREATE PROGRAM explorer_23_ex_11

!prompts
PROMPT "Output to File/Printer/MINE" = MINE

!report prolog
;Request HNAM sign-on when executed from CCL on host
IF (VALIDATE(IsOdbc, 0) = 0) EXECUTE CCLSECLOGIN ENDIF

free record per_ord
```

34TRGINT000008

Owner: Product Specialist Team Lead

Reviewed By: Mike Schweder

Approved By: Bob Ross

```
record per_ord (
    1 person[*]
        2 name      = c25
        2 orders[*]
            3 id       = f8
            3 mne     = c25  )

!set maximum execute time for query/report
SET MaxSecs = 0
IF (VALIDATE(IsOdbc, 0) = 1)  SET MaxSecs = 80  ENDIF
!fields
SELECT    INTO $1
    P.PERSON_ID,
    P.NAME_FULL_FORMATTED,
    O.ORDER_ID,
    O.ORDER_MNEMONIC

!tables
FROM    PERSON  P,
        ORDERS  O

!qualifications
plan p where  p.person_id between 150 and 250
JOIN o  WHERE  p.person_id = o.person_id

!Sorting
ORDER BY      P.PERSON_ID,
               O.ORDER_ID

Head Report
cnt = 0
cntx = 0
cntp = 0
cntpx = 0
cnto = 0

Head P.PERSON_ID
cntp = cntp+1
if(mod(cntp,10) = 1)
    stat = alterlist(per_ord->person,cntp+9)
endif
per_ord->person[cntp]->name =  substring(1,25, p.name_full_formatted)
;initialize the orders segment and cnt variable
cnt = 0
ROW + 0

Detail
cnt = cnt+1
if(mod(cnt,10) = 1)
    stat = alterlist(per_ord->person[cntp]->orders,cnt+9)
endif
per_ord->person[cntp]->orders[cnt]->id = o.order_id
per_ord->person[cntp]->orders[cnt]->mne =
```

34TRGINT000008

Owner: Product Specialist Team Lead

Reviewed By: Mike Schweder

Approved By: Bob Ross

```
substring(1,25,o.order_mnemonic)

Foot P.PERSON_ID
    stat = alterlist(per_ord->person[cntp]->orders,cnt)
    ROW + 0

Foot Report
    stat = alterlist(per_ord->person,cntp)
    for(cntp = 1 to cntp)
        col 0 per_ord->person[cntp]->name
        cnto = size(per_ord->person[cntp]->orders,5)
        row +1
        for (cntx = 1 to cnto)
            col 10 per_ord->person[cntp]->orders[cntx]->id
            col +2 per_ord->person[cntp]->orders[cntx]->mne
            row +1
        endfor
    endfor

!control options
WITH MAXREC = 20, TIME = VALUE( MaxSecs ), NOHEADING, FORMAT= VARIABLE

END
GO
```

Exercise 12

```
DROP PROGRAM explorer23_class_ex12 GO
CREATE PROGRAM explorer23_class_ex12

!prompts
PROMPT "Output to File/Printer/MINE" = MINE

!report prolog
;Request HNAM sign-on when executed from CCL on host
IF (VALIDATE(IsOdbc, 0) = 0) EXECUTE CCLSECLOGIN ENDIF

SELECT INTO TABLE CCLCUSTTABLE
    P.PERSON_ID,
    P.NAME_LAST_KEY
FROM PERSON_P
WHERE P.NAME_LAST_KEY = "R*"
ORDER P.PERSON_ID
WITH MAXREC = 100,
    ORGANIZATION = I

!set maximum execute time for query/report
SET MaxSecs = 0
IF (VALIDATE(IsOdbc, 0) = 1) SET MaxSecs = 15 ENDIF

!challenge steps
!fields
SELECT INTO $1
    O.ORDER_MNEMONIC,
```

34TRGINT000008

Owner: Product Specialist Team Lead

Reviewed By: Mike Schweder

Approved By: Bob Ross

```
C.PERSON_ID,  
C.NAME_LAST_KEY  
  
!tables  
FROM ORDERS O,  
CCLCUSTTABLE C  
  
!qualifications  
PLAN O  
JOIN C WHERE O.PERSON_ID = C.PERSON_ID  
  
!control options  
WITH MAXREC = 100, TIME = VALUE( MaxSecs ), FORMAT, SKIPREPORT= 1  
  
END  
GO
```

Exercise 13

```
DROP PROGRAM explorer23_class_ex13 GO  
CREATE PROGRAM explorer23_class_ex13  
  
!prompts  
PROMPT "Output to File/Printer/MINE" = MINE  
  
!report prolog  
;Request HNAM sign-on when executed from CCL on host  
IF (VALIDATE(IsOdbc, 0) = 0) EXECUTE CCLSECLOGIN ENDIF  
  
DECLARE MRN_VAR = F8  
SET MRN_VAR = UAR_GET_CODE_BY("MEANING", 4, "MRN")  
  
SET MaxSecs = 0  
IF (VALIDATE(IsOdbc, 0) = 1) SET MaxSecs = 15 ENDIF  
  
!fields  
SELECT INTO $1  
P.PERSON_ID,  
NAME = SUBSTRING(1,20,P.NAME_FULL_FORMATTED),  
SIDE = DECODE(A.SEQ,"ADDR ", PA.SEQ, "ALIAS"),  
A.ADDRESS_ID,  
ADDR = SUBSTRING(1,20,A.STREET_ADDR),  
PA.PERSON_ALIAS_ID,  
ALIAS = SUBSTRING(1,20,PA.ALIAS)  
  
!tables  
FROM PERSON P,  
ADDRESS A,  
PERSON_ALIAS PA,  
DUMMYT D1,  
DUMMYT D2  
  
!qualifications  
PLAN P WHERE P.PERSON_ID>0  
JOIN D1
```

34TRGINT000008

Owner: Product Specialist Team Lead

Reviewed By: Mike Schweder

Approved By: Bob Ross

```
JOIN (A WHERE A.PARENT_ENTITY_ID = P.PERSON_ID AND
          A.PARENT_ENTITY_NAME = "PERSON")
ORJOIN D2
JOIN (PA WHERE PA.PERSON_ID = P.PERSON_ID)

!control options
WITH MAXREC = 1000, TIME = VALUE( MaxSecs ), FORMAT, SKIPREPORT= 1

END
GO
```

Exercise 14

```
DROP PROGRAM explorer23_class_ex7 GO
CREATE PROGRAM explorer23_class_ex7

!prompts
PROMPT "Output to File/Printer/MINE" = MINE

!report prolog
;Request HNAM sign-on when executed from CCL on host
IF (VALIDATE(IsOdbc, 0) = 0) EXECUTE CCLSECLOGIN ENDIF

SET MaxSecs = 0
IF (VALIDATE(IsOdbc, 0) = 1) SET MaxSecs = 15 ENDIF

!fields
SELECT INTO $1
  E.ENCNTR_ID,
  E.ENCNTR_TYPE_CLASS_DISP = UAR_GET_CODE_DISPLAY(E.ENCNTR_TYPE_CLASS_CD),
  SIDE = decode(O.SEQ, "ORD", EP.SEQ, "EPR"),
  O.ORDER_ID,
  O.ORDER_MNEMONIC,
  EP.ENCNTR_PRSNL_R_DISP = UAR_GET_CODE_DISPLAY(EP.ENCNTR_PRSNL_R_CD),
  P.NAME_FULL_FORMATTED

!tables
FROM ENCOUNTER E,
  ORDERS O,
  ENCNTR_PRSNL_RELTN EP,
  PRSNL P,
  DUMMYT D1,
  DUMMYT D2

!qualifications
PLAN E WHERE E.ENCNTR_ID != 0.0
JOIN D1
JOIN (O WHERE E.ENCNTR_ID = O.ENCNTR_ID)
ORJOIN D2
JOIN (EP WHERE E.ENCNTR_ID = EP.ENCNTR_ID
JOIN P WHERE EP.PRSNL_PERSON_ID = P.PERSON_ID)
ORDER BY E.ENCNTR_ID,
          SIDE DESC
```

Revision: 002
Eff. Date: August 16, 2005

Millennium Explorer 2-3 Answers

Page 15 of 15

34TRGINT000008

Owner: Product Specialist Team Lead

Reviewed By: Mike Schweder

Approved By: Bob Ross

```
!control options
WITH MAXREC = 100, TIME = VALUE( MaxSecs ), FORMAT, SKIPREPORT= 1
```

```
END
GO
```

